


**SONY®**

Sony España, S.A.  
Sabino de Arana, 42-44  
08028 Barcelona  
España



PDV  
FOM: Marketing Y Publicidad Directe S.A.  
Apartado de Correos 61301  
28030 MADRID  
TARIFA REDUCIDA CLASE 29

17121  
MADRID  
-9.6.87

ESPAÑA  
10.50



JOSE ENRIQUE [REDACTED]

[REDACTED]  
ERA ALTA  
MURCIA

*Estimado JOSE ENRIQUE:*

*Bienvenido al Club SONY MSX. Nuestro Club está formado por todos los que, como tú, nos interesamos en estar bien informados y en sacar el mayor partido de nuestros ordenadores MSX.*

*A partir de ahora, como socio del Club, recibirás cada mes las noticias más frescas, la más amplia información sobre las últimas novedades y todo aquello que más te interesa para que tu MSX funcione a toda marcha.*

*Además, para que amplíes tu programoteca, el Club prepara y distribuye gratuitamente entre sus socios muchos programas.*

*Te enviamos el primero: "Policia y ladrón", un apasionante juego que tu mismo puedes cargar en tu ordenador. Y para que la presentación te quede perfecta, incluimos también una divertida carátula a todo color.*

*Nosotros ya nos hemos presentado. Ahora nos gustaría conocer tus intereses y preferencias de cara a nuestros próximos envíos. Para ello, rellena, por favor, el cuestionario que te adjuntamos con esta carta y envíanoslo antes de 15 días en el sobre que te incluimos.*

*Y, de momento, nada más. Esperamos que la revista y el programa te gusten. Y ...no olvides el cuestionario.*

*Hasta el próximo mes.*



LUIS CARRERAS

SONY

**N1** POLICIA Y LADRON **CLUB SONY MSX**

**POLICIA Y LADRON**



## POLICÍA Y LADRÓN

Programa de entretenimiento donde debes poner a prueba tu habilidad y razonamiento.

Dispones de 6 policías para acorralar al ladrón.

Solo pueden desplazarse según el «salto del caballo» de ajedrez.

Debes acorralar al ladrón e impedir que pueda moverse, en el menor número de movimientos posibles.

No puedes retroceder.



INSTRUCCIONES: Una vez has introducido el programa al ordenador, ejecútalo pulsando la tecla «F5».

Para mover uno de los seis policías debes escoger uno de ellos (letra A hasta F) y pulsar la tecla correspondiente.

El ordenador te señalará, con un número, las posibles casillas donde puedes ir. Escoge uno de ellos, pulsando el número correspondiente.

Alternativamente irá moviéndose la policía y el ladrón hasta que consigas acorralarlo.

Debes tener presente que ni la policía ni el ladrón pueden retroceder a la casilla de donde provenía.

Si deseas salvar el programa: debes pulsar simultáneamente la tecla CTRL y STOP e introducir la siguiente instrucción:

1. Pulsar simultáneamente las teclas SHIFT y «F1».

2. Si es cassette: CSAVE "POLICIA"

Si es unidad de diskette: SAVE "POLICIA"

En caso de errores, o si la pantalla se queda de color blanco, debes pulsar simultáneamente las teclas «SHIFT» y «F1» y teclear de nuevo la línea que se especifica:

SYNTAX ERROR N \_\_\_\_\_

Una vez rectificado, pulsar de nuevo la tecla «F5».

Te adjuntamos también una carátula del programa para que empieces a crear tu «programoteca».

NOTA: En algunos ordenadores la tecla «SHIFT» se simboliza con una flecha que apunta hacia arriba.

INSTRUCCIONES DE CARGA: Si deseas cargar el programa al ordenador, introduce la siguiente instrucción:

Si es cassette: CARD

Si es de Unidad de diskette: LOAD "POLICIA"

```
10 '*****
20 'POLICIA Y LADRON
30 '*****
40 OPEN"GRP:" FOR OUTPUT AS #1
50 DIM B(7,7),P(8),Q(8)
60 COLOR 15,15,7
70 SCREEN 2,2
80 FOR I=1 TO 5
90 FOR J=1 TO 5
100 B(I,J)=1
110 NEXT:NEXT
120 FOR I=1 TO 5
130 FOR J=1 TO 5
140 LINE(I*32-15,J*32-15)-(I*32+29
-15,J*32+29-15),4,BF
150 NEXT:NEXT
160 LINE(180,17)-(245,100),4,BF
170 GOSUB 1610
180 GOSUB 1630
190 DEF FNZX(X)=X*5+185
200 DEF FNZY(Y)=Y*48+22
210 X=0:Y=0:GOSUB 1470
220 DRAW"BM198,44"
230 PRINT #1,"OLICIA"
240 X=0:Y=1:GOSUB 1560
250 DRAW"BM208,83"
260 PRINT #1,"ADRON"
270 DRAW"BM215,62"
280 PRINT #1,"Y"
290 DEF FNZX(X)=X*32+3-15
300 DEF FNZY(Y)=Y*32+1-15
310 FOR I=1 TO 6
320 READ XT(I),YT(I)
330 X=XT(I):Y=YT(I):GOSUB 1470
340 PSET(FNZX(X)+20,FNZY(Y)+20),4
350 PRINT #1,CHR$(64+I)
360 B(XT(I),YT(I))=4
370 NEXT
380 DATA 1,1,3,1,5,1,1,5,3,5,5,5
390 B(3,3)=2:XJ=3:YJ=3
400 X=XJ:Y=YJ:GOSUB 1560
410 '*****
420 GOSUB 1610
430 DRAW"BM183,133"
440 PRINT #1,"TIRA EL"
450 DRAW"BM183,143"
460 PRINT #1,"LADRON"
470 X=XJ:Y=YJ:GOSUB 1280
480 TT=0
490 FOR I=1 TO CO
500 IF B(P(I),Q(I))=4 THEN 540
510 IF B(P(I),Q(I))=3 THEN B(P(I),
Q(I))=1:GOTO 540
520 TT=TT+1
530 P1(TT)=P(I):Q1(TT)=Q(I)
540 NEXT
550 IF TT=1 THEN R=1:GOTO 660
560 IF TT=0 THEN 1730
570 I=1
580 X=P1(I):Y=Q1(I):GOSUB 1280
590 U=0:FOR J=1 TO CO
600 IF B(P(J),Q(J))=1 THEN U=U+1
610 NEXT J
620 IF U>1 THEN R=I:GOTO 660
630 IF I<>TT THEN I=I+1:GOTO 580
640 R=1
650 '*****
660 B(XJ,YJ)=1
670 X=XJ:Y=YJ:GOSUB 1440
680 B(XJ,YJ)=3
690 XJ=P1(R):YJ=Q1(R)
700 B(XJ,YJ)=2
710 X=XJ:Y=YJ:GOSUB 1560
720 TE=TE+1
730 GOSUB 1630
740 DRAW"BM185,112"
750 PRINT#1,"MOV.: ";TE
```

```

760 '***** Tirada
770 GOSUB 1610
780 DRAW"BM184,133"
790 PRINT #1,"TIRA LA"
800 DRAW"BM183,143"
810 PRINT #1,"POLICIA"
820 DRAW"BM183,153"
830 PRINT #1,"(A-F)?"
840 A$=INKEY$:IF A$="" THEN 840
850 IF A$="A" OR A$="a" THEN K=1:G
OTO 920
860 IF A$="B" OR A$="b" THEN K=2:G
OTO 920
870 IF A$="C" OR A$="c" THEN K=3:G
OTO 920
880 IF A$="D" OR A$="d" THEN K=4:G
OTO 920
890 IF A$="E" OR A$="e" THEN K=5:G
OTO 920
900 IF A$="F" OR A$="f" THEN K=6:G
OTO 920
910 GOTO 840
920 DRAW"BM233,153"
930 PRINT #1,A$
940 X=XT(K):Y=YT(K):GOSUB 1280
950 TT=0
960 FOR I=1 TO CO
970 IF B(P(I),Q(I))=4 OR B(P(I),Q(
I))=2 THEN 1030
980 IF P(I)=XD(K) AND Q(I)=YD(K) T
HEN 1030
990 PSET(FNZX(P(I))+2,FNZY(Q(I))+1
),4
1000 TT=TT+1
1010 PRINT #1,TT
1020 P(TT)=P(I):Q(TT)=Q(I)
1030 NEXT
1040 IF TT=0 THEN 1660
1050 XD(K)=XT(K):YD(K)=YT(K)
1060 DRAW"BM183,163"
1070 TT$=RIGHT$(STR$(TT),1)
1080 PRINT #1,"(1-";TT$;")/"
1090 A$=INKEY$:IF A$="" THEN 1090
1100 IF VAL(A$)<1 OR VAL(A$)>TT TH
EN 1090
1110 A=VAL(A$)
1120 B(XT(K),YT(K))=1
1130 FOR I=1 TO TT
1140 X=P(I):Y=Q(I):GOSUB 1440
1150 NEXT I
1160 X=XT(K):Y=YT(K):GOSUB 1440
1170 XT(K)=P(A):YT(K)=Q(A)
1180 X=XT(K):Y=YT(K):GOSUB 1470
1190 PSET(FNZX(X)+20,FNZY(Y)+20),4
1200 PRINT #1,CHR$(64+K)
1210 B(XT(K),YT(K))=4
1220 TE=TE+1
1230 GOSUB 1630
1240 DRAW"BM185,112"
1250 PRINT #1,"MOV.:";TE
1260 GOTO 420
1270 '*****
1280 CO=0
1290 Y1=Y-2:X1=X+1:GOSUB 1380
1300 Y1=Y-1:X1=X+2:GOSUB 1380
1310 Y1=Y+1:X1=X+2:GOSUB 1380
1320 Y1=Y+2:X1=X+1:GOSUB 1380
1330 Y1=Y+2:X1=X-1:GOSUB 1380
1340 Y1=Y+1:X1=X-2:GOSUB 1380
1350 Y1=Y-1:X1=X-2:GOSUB 1380
1360 Y1=Y-2:X1=X-1:GOSUB 1380
1370 RETURN
1380 IF X1<0 OR Y1<0 THEN RETURN
1390 IF B(X1,Y1)=0 THEN RETURN
1400 CO=CO+1
1410 P(CO)=X1:Q(CO)=Y1
1420 RETURN
1430 '*****
1440 LINE(FNZX(X),FNZY(Y))-(FNZX(X
)+26,FNZY(Y)+27),4,BF
1450 RETURN
1460 '***** Letra "P"
1470 BEEP
1480 LINE(FNZX(X),FNZY(Y))-(FNZX(X
)+5,FNZY(Y)+27),15,BF
1490 LINE(FNZX(X),FNZY(Y))-(FNZX(X
)+15,FNZY(Y)+5),15,BF
1500 LINE(FNZX(X),FNZY(Y)+13)-(FNZ
X(X)+15,FNZY(Y)+18),15,BF
1510 CIRCLE(FNZX(X)+15,FNZY(Y)+9),
4,15,4.9,1.6
1520 CIRCLE(FNZX(X)+15,FNZY(Y)+9),
9,15,4.9,1.6
1530 PAINT(FNZX(X)+20,FNZY(Y)+9),1
5,15
1540 RETURN
1550 '***** Letra "L"
1560 BEEP
1570 LINE(FNZX(X),FNZY(Y)+22)-(FNZ
X(X)+23,FNZY(Y)+27),15,BF
1580 LINE(FNZX(X),FNZY(Y))-(FNZX(X
)+5,FNZY(Y)+27),15,BF
1590 RETURN
1600 '*****
1610 LINE(180,130)-(245,174),4,BF
1620 RETURN
1630 LINE(180,105)-(247,125),4,BF
1640 RETURN
1650 '*****
1660 GOSUB 1610
1670 DRAW"BM183,145"
1680 PRINT #1,"NO PUEDO"
1690 FOR J=1 TO 5
1700 BEEP:NEXT J
1710 GOTO 770
1720 '***** Final
1730 GOSUB 1610
1740 DRAW"BM182,138"
1750 PRINT #1,"BRAVO !!"
1780 DRAW"BM183,158"
1790 PRINT #1,"MUY BIEN"
1800 X$=INKEY$:IF X$="" THEN 1800
1810 CLOSE:CLEAR:GOTO 10

```